

Independent Study: Tool-Assisted Verification of C Code using Floyd-Hoare Logic

Nikolas Mählmann

April 9, 2020

1 Introduction

This document explains the usage and inner workings of `verify-c`. `verify-c` is a command line tool which verifies programs written in a subset of C. The tool is based on Floyd-Hoare logic, which was intensively discussed in the course *Korrekte Software: Grundlagen und Methoden* at the University of Bremen held by Serge Autexier and Christoph Lüth in the summer semester 2019. `verify-c` is an educational implementation of the discussed techniques aimed to deepen their understanding and explore which challenges arise when formally verifying software. `verify-c` is written in Haskell and the source code is available online at <https://github.com/nmaehlmann/verify-c>.

2 Installation

`verify-c` can be built using the Haskell build tool `stack` by calling:

```
stack install
```

in the root directory. Additionally `verify-c` relies on the Z3 theorem prover which has to be installed and added to the `PATH` variable. It can be downloaded at <https://github.com/Z3Prover/z3>.

3 Usage

`verify-c` parses program code written in a subset of C. Each function is annotated with logical pre- and postconditions, which specify the contract of

the function. Based on the parsed program it generates a set of verification conditions. The verification conditions are exported to a theorem prover, which checks whether or not they are satisfied. If all verification conditions are proven successfully, the implemented functions satisfy their contract. Lets start with a simple example. Listing 1 shows a verified implementation of the faculty function.

Listing 1: faculty.c0

```
1 int faculty(int n){
2     precondition("n >= 0");
3     postcondition("\result == fac(n)");
4
5     p = 1;
6     c = 1;
7     while(c <= n){
8         invariant("p == fac(c - 1) && c <= n + 1 && c > 0");
9         p = p * c;
10        c = c + 1;
11    }
12    return p;
13 }
```

It is written regular C code but additional function calls have been added to specify the contract of the function in order to verify it. The precondition in l.2 states that the function argument **n** has to be positive. The postcondition in l.3 that after calling the function will return **fac** of **n**. When verified successfully these conditions have the following semantic: If the function **faculty** is called with a positive argument **n** and it terminates, then the return value of this function will equal **fac** of **n**. Furthermore the while loop is annotated with an invariant (l.8). The invariant has to be satisfied before the while loop is entered as well as after each loop is completed. The specification of preconditions, postconditions and invariants is mandatory and missing specifications will result in a parser error.

Additionally to the C source code **verify-c** requires an environment file if custom functions or predicates are used in specifications. The function **fac** used in the precondition and in the invariant is such a custom function. It is specified in Listing 2.

Listing 2: faculty.env

```
1 (declare-fun fac (Int) Int)
2 (assert (= (fac 0) 1))
3 (assert (forall ((nn Int)) (implies (< 0 nn) (= (fac nn) (* nn
    ↪ (fac (- nn 1)))))))
```

The specification of the environment is written in the SMT-LIB format and verbatim fed into the Z3 prover. More information regarding the SMT-LIB language can be found online at <http://smtlib.cs.uiowa.edu/language.shtml>. In order to be found by `verify-c` the environment has to have the same name as the source file but with an `.env` extension. With the environment in place the source code of the faculty function can now be verified by calling:

```
verify-c faculty.c0
```

which produces the following output:

```
Generated 3 verification condition(s). Starting proof:
[1/3] : Precondition faculty : OK
[2/3] : While Case True (1:8) : OK
[3/3] : While Case False (1:8) : OK

Summary: VERIFICATION OK
```

Hooray! Three verification conditions were generated by `verify-c` and successfully proven by Z3. One originates from the precondition of the faculty function, two from the invariant of the while loop.

In this case every verification condition could be proven, which is indicated by the status code OK. Other status codes are:

- **SIMPLIFY FAILED:** The verification conditions could not be simplified enough to be proven. This is most likely caused by ambiguous dereferencing.
- **SMT EXPORT FAILED:** The verification condition could not be translated into SMT-LIB code. This is most likely caused by ambiguous referencing.
- **VIOLATED:** The verification condition was disproven. The specification and program do not match.

- **TIMEOUT:** Z3 timed out while trying to prove the verification condition. It could neither disprove nor prove it.
- **SMT ERROR:** Z3 produced an unknown error.
- **SKIPPED:** The verification condition was skipped because of a previous error.

The generated verification conditions and SMT-LIB code as well as logfiles are stored in the `.\target` folder created by `verify-c`. This is the place to look at in case verification fails.

`verify-c` can be further configured by using command line options. A list of all available can be displayed by calling:

```
verify-c -h
```

which outputs:

```
Help Options:
-h, --help
    Show option summary.
--help-all
    Show all help options.

Application Options:
--color :: bool
    Whether or not to use ANSI colors.
    default: false
--timeout :: int
    SMT solver timeout in seconds.
    default: 5
--no-skip :: bool
    Whether or not to continue verification after a condition
    ↪ could not be
    verified.
    default: false
```

4 Implementation

`verify-c` is written in Haskell and the source code is available online at <https://github.com/nmaehlmann/verify-c>.

4.1 Parsing

Parsing of the source code is done using the parser combinators library `parsec`. This is a standard procedure so I will not go into further details about parsing. The result of the parsing process is an Abstract Syntax Tree (AST) which is annotated with first order logic formulas.

4.2 Logical Formulas

Logical formulas are the core data structure on which most of the verification logic operates. They are implemented by the GADT `BExp` shown in Listing 3.

Listing 3: `BExp`

```
1 data BExp l m where
2   BTrue  :: BExp l m
3   BFalse :: BExp l m
4   BNeg   :: BExp l m -> BExp l m
5   BBinExp :: BBinOp -> BExp l m -> BExp l m -> BExp l m
6   BComp   :: CompOp -> AExp l m -> AExp l m -> BExp l m
7   BForall :: Idt -> BExp FO m -> BExp FO m
8   BExists :: Idt -> BExp FO m -> BExp FO m
9   BPredicate :: Idt -> [AExp FO m] -> BExp FO m
```

The `BExp` type is parameterized by two arguments `l` and `m`.

The first argument `l` characterizes the type of logic that is used. It can take two values:

1. `C0`: The logical operations that can be used as a part of the C programming language for example to formulate the condition of a while loop.
2. `F0`: First order logic which is used to specify preconditions, postconditions and invariants.

While the `true` and `false` constants, negations, boolean operators, and comparisons are available in every logic, quantifiers and predicates are only available in first order logic. This is guaranteed by the type system through the usage of GADTs. Since `BExp C0 m` is a subset of `BExp F0 m`, the first is converted to the latter during the actual generation of verification conditions.

The second argument of `BExp` is `m` which characterizes the memory model that is used. It can also take two values:

1. **Plain:** A user facing symbolic memory model that is used during the development of the program and the specification.
2. **Refs:** An axiomatic memory model which is used internally during verification condition generation in order to support references.

4.3 Arithmetic Expressions

Arithmetic expressions are expressions which evaluate to an integer. They are used on the right hand side of assignments, as array indices or as a part of a comparison operation in a logical formula. Arithmetic expressions are modelled by the `AExp` GADT which is shown in Listing 4.

Listing 4: `AExp`

```

1 data AExp l m where
2   ALit :: Integer -> AExp l m
3   AIdt :: LExp l m -> AExp l m
4   ABinExp :: ABinOp -> AExp l m -> AExp l m -> AExp l m
5   AFunCall :: Idt -> [AExp FO m] -> AExp FO m
6   ALogVar :: Idt -> AExp FO m
7   AAddress :: LExp l Plain -> AExp l Plain

```

As parts of `BExps` they are also parameterized with the type of logic and memory model. Integer literals, variable names and binary calculations are supported for every combination of parameters. Logical variables and function calls as parts of an `AExp` require first order expressions, so they can only be used in for specification purposes. As part of the C program code function calls do not form an `AExp` but are treated as a separate statement. This limitation is further explained in section ???. The address operator (`&`) is transformed into an `LExpression` in the `Refs` memory model so it is only available in the `Plain` memory model.

4.4 LExpressions

`LExpressions` are expressions which can be used on the left hand side of assignments or as variable identifiers as parts of arithmetic expressions. `LExpressions` are modelled by the `LExp` GADT which is shown in Listing 5.

Listing 5: LExp

```

1 data LExp l m where
2   LIdt :: Idt -> LExp l m
3   LArray :: LExp l m -> AExp l m -> LExp l m
4   LStructurePart :: LExp l m -> Idt -> LExp l m
5   LRead :: State -> LExp l Refs -> LExp l Refs
6   LDeref :: LExp l Plain -> LExp l Plain

```

As parts of **AExps** they are also parameterized with the type of logic and memory model. Identifiers and array and struct accessors are available for every combination of parameters. Similar to the address operator, the dereferencing operator (*****) is only available in the **Plain** memory model. The **LRead LExp** is the core of the **Refs** memory model on which will be explained in more detail in the next section.

4.5 Memory Models

In the symbolic memory model **C0** each LExpression is assigned a value. To verify a program using references however, it is necessary to transform the symbolic model into an axiomatic one. In the axiomatic model, each LExpression (except **LRead**) is assigned a memory address. The actual value is obtained by looking up the memory address in a program state:

$$read(\sigma, l)$$

Reads are modelled by the **LRead** LExpression. Assigning a value to an address is creates an updated state:

$$\sigma_2 = update(\sigma_1, l, v)$$

This is modelled by the **State** type shown in Listing 6.

Listing 6: State

```

1 data State
2   = Atomic String
3   | Update State (LExp F0 Refs) (AExp F0 Refs)

```

Using this axiomatic model, LExpressions, Referencing and Dereferncng can be treated uniformly:

$$\mathbf{a} \hat{=} read(\sigma, a)$$

$$\&\mathbf{a} \hat{=} a$$

$$*\mathbf{a} \hat{=} read(\sigma, read(\sigma, a))$$

Often programs contain multiple assignments which leads to deeply nested states, for example:

$$update(update(update(\sigma, l_1, v_1), l_2, v_2), l_3, v_3)$$

The situation gets worse, when references are involved because dereferencing an LExpression doubles the amount of states. To keep the states small the following simplification rules are introduced:

$$l_1 = l_2 \Rightarrow update(update(\sigma, l_2, v_2), l_1, v_1) = update(\sigma, l_1, v_1)$$

$$l_1 = l_2 \Rightarrow read(update(\sigma, l_2, v), l_1) = v$$

$$l_1 \neq l_2 \Rightarrow read(update(\sigma, l_2, v), l_1) = read(\sigma, l_1, v)$$

To apply these simplifications it is crucial to decide whether or not two LExpressions are equal. This however is not always possible. Two references might point to the same address, or two array indices might have the same value:

$$*\mathbf{a} \stackrel{?}{=} *\mathbf{b}$$

$$\mathbf{a}[\mathbf{i}] \stackrel{?}{=} \mathbf{a}[\mathbf{j}]$$

The following heuristic comparison algorithm was implemented in the module `Memory.Eq`:

$$\begin{aligned}
& cmp(a, a) = Eq \\
& cmp(a, b) = NotEq \quad \text{if } a \neq b \text{ is predefined} \\
& cmp(a, b) = NotEq \quad \text{if } a \text{ was just initialized} \\
& cmp(a, b) = NotEq \quad \text{if } b \text{ was just initialized} \\
& cmp(read(\sigma, a), read(\sigma, a)) = cmp(a, b) \\
& cmp(read(\sigma, a), b) = Undecidable \\
& cmp(a, read(\sigma, b)) = Undecidable \\
& cmp(a.i, b.j) = cmp(a, b) \quad \text{if } i = j \\
& cmp(a.i, b.j) = NotEq \quad \text{if } i \neq j \\
& cmp(a[i], b[j]) = Eq \quad \text{if } cmp(a, b) = Eq \wedge cmpA(i, j) = Eq \\
& cmp(a[i], b[j]) = NotEq \quad \text{if } cmp(a, b) = NotEq \vee cmpA(i, j) = NotEq \\
& cmp(a[i], b[j]) = Undecidable \quad \text{otherwise} \\
& cmp(a, b) = NotEq \quad \text{otherwise} \\
\\
& cmpA(a, a) = Eq \\
& cmpA(a, b) = NotEq \quad \text{if } a, b \text{ are both literals} \\
& cmpA(a, b) = Eq \quad \text{if } a, b \text{ are both identifiers} \wedge cmp(a, b) = Eq \\
& cmpA(a, b) = Undecidable \quad \text{otherwise}
\end{aligned}$$

cmp compares two LExpressions and should return *Eq* if two *LExpression* evaluate to the same memory address. *cmpA* is used to compare array indices and should return *Eq* if two arithmetic expressions evaluate to the same integer. $a \neq b$ is a predefined inequality in the right hand side of an implication if $a \neq b$ is true in the left hand side of that implication. a was just initialized if the previous statement is the declaration of a . In this case it is assumed, that the operating system assigned a fresh memory address to a . Both, the predefined inequalities and the set of just initialized identifiers have to be passed to the functions as a context.

4.6 Simplification

With the simplification rules and comparison function in place the simplification algorithm can be implemented. Applying one simplification rule to

a logical formula can lead to the opportunity to apply another one, leading to an expression collapsing step by step. Therefore the simplification algorithm has to run repeatedly until no further simplifications are possible. To conveniently keep track whether a simplification has happened, or the result remained unchanged a custom `Updated` monad shown in 7 is introduced:

Listing 7: the `Updated` monad

```

1 data Updated a = Updated a | Unchanged a
2
3 instance Monad Updated where
4     return a = Unchanged a
5     (Updated a) >>= f = Updated $ unwrap $ f a
6     (Unchanged a) >>= f = f a
7
8 unwrap :: Updated a -> a
9 unwrap (Updated a) = a
10 unwrap (Unchanged a) = a

```

If an `Updated` value is composed, the result is also `Updated`. As previously presented, the comparison algorithm for `LExpressions` requires a context in which predefined inequalities and local variables can be looked up. This context is made accessible by wrapping the `Updated` monad into a `Reader` monad, which carries the required information. The obtained nested monad is aliased as `Simplified` and presented in Listing 8.

Listing 8: the `Simplified` monad

```

1 type Simplified = ReaderT SimplificationCtx Updated
2
3 data SimplificationCtx = SimplificationCtx
4     { inequalities :: Set Inequality
5     , localVars :: Set (LExp F0 Refs)
6     }
7
8 type Inequality = Set (LExp F0 Refs)
9
10 update :: a -> Simplified a
11 update a = lift $ Updated a

```

The simplification algorithm can now be implemented as a set of func-

tions which recursively traverse and simplify a BExp. It is located in the `Logic.Simplification` module. The implementation of each of the three simplification rules is shown in Listing 9

Listing 9: implementation of the simplification rules

```

1 simplifyState :: State -> Simplified State
2 simplifyState original@(Update (Update s l1 _) l2 w) = do
3   memComparison <- compareLEExp l1 l2
4   case memComparison of
5     MemEq -> update $ Update s l2 w
6     _ -> simplifyState' original
7 simplifyState s = simplifyState' s
8
9 simplifyAExp :: AExpFO -> Simplified AExpFO
10 simplifyAExp original@(AIdt (LRead (Update state toUpdate aExp)
11   ↪ toRead)) = do
12   memComparison <- compareLEExp toRead toUpdate
13   case memComparison of
14     MemEq -> update aExp
15     _ -> simplifyAExp' original
16 simplifyAExp a = simplifyAExp' a
17
18 simplifyLEExp :: LExpFO -> Simplified LExpFO
19 simplifyLEExp original@(LRead (Update state toUpdate _) toRead)
20   ↪ = do
21   memComparison <- compareLEExp toRead toUpdate
22   case memComparison of
23     MemNotEq -> update $ LRead state toRead
24     _ -> simplifyLEExp' original
25 simplifyLEExp l = simplifyLEExp' l

```

`simplifyState'`, `simplifyAExp'`, and `simplifyLEExp'` are not shown in the listing, as they just recursively descend the expression. The corresponding function to `simplifyBExp`, which starts simplification on the formula level has a special handling for implication. Inequalities specified on the left hand side of an implication can be used to simplify the right hand side of the same implication as described in section 4.5. For this purpose the context used to simplify the right hand side is enriched with the inequalities that were found

in the left hand side, which is depicted in Listing 10.

Listing 10: searching for predefined inequalities

```

1 simplifyBExp :: BExpF0 -> Simplified BExpF0
2 simplifyBExp (BBinExp op l r) = do
3   updatedL <- simplifyBExp l
4   let lhsInequalities = if op == Implies
5     then findInequalities updatedL
6     else Set.empty
7   updatedR <- local (addInequalities lhsInequalities) $
8     ↪ simplifyBExp r
9   return $ BBinExp op updatedL updatedR
10 ... other cases of simplifyBExp: recursively simplify BExp ...
11 findInequalities :: BExpF0 -> Set Inequality
12 findInequalities (BComp NotEqual (AIdt l1) (AIdt l2)) =
13   Set.singleton $ notEqual l1 l2
14 findInequalities (BBinExp And fo1 fo2) = Set.union (
15   ↪ findInequalities fo1) (findInequalities fo2)
16 findInequalities _ = Set.empty

```

The search for inequalities is again implemented as a simple heuristic only. Inequalities are found if they are either specified on the top level of the formula, or are part of a (possibly nested) conjunction.

4.7 Statements

The program that should be verified into a sequence of statements. The type `Stmt` which models a statement in `verify-c` is depicted in Listing 11.

Listing 11: Stmt

```
1 type BExp' l = BExp l Plain
2 type AExp' l = AExp l Plain
3 type LExp' l = LExp l Plain
4
5 data Stmt
6   = Empty
7   | Declaration Idt
8   | Assignment (LExp' C0) (AExp' C0)
9   | While (BExp' C0) (BExp' F0) Stmt LineNo
10  | ITE (BExp' C0) Stmt Stmt
11  | FunCall (Maybe (LExp' C0)) Idt [AExp' C0] LineNo
12  | Assertion (BExp' F0) LineNo
13  | Return (Maybe (AExp' C0))
14  | Seq Stmt Stmt
```

Since the statements are obtained directly from the parser, all statement use the Plain memory model. `verify-c` supports the following statements:

Empty

Empty statements have no effect.

Declarations

Declarations declare the memory address of an identifiers as fresh.

Assignments

Assignments which also act as declarations like `int i = 0;` are split into a declaration and an assignment by the parser.

While Loops

While loops consist of a C0 logic condition, an F0 logic invariant and a body. The current line number is included for logging purposes.

If-Then-Else Branches

If-Then-Else branches consist of a C0 logic condition, a true branch, and a false branch. If conditions without an else branch are parsed as If-Then-Else branches with an empty false branch.

Function Calls

Function Calls consisting of the variable the result of the function is assigned to, the function name, and a list of function arguments. The current line number is included for logging purposes. Since some functions do not return a result, the assigned variable is optional. Function calls are modelled as separate statements instead of an arithmetic expressions in order to avoid dealing with ambiguous orders of side effects in expressions like `n = inc(&n) + inc(&n);`. Function calls which also act as declarations are split into a declaration and a function call by the parser.

Assertions

Assertions modify the current precondition.

Return Statements

Return statements which may or may not return a value.

Sequences

Statements can be combined into sequences.

4.8 Applying the Floyd-Hoare Logic

The Floyd-Hoare logic was intensively discussed in the *Korrekte Software: Grundlagen und Methoden* course. Given a postcondition, for every statement an approximate weakest precondition (*awp*) is calculated. Traversing the statements in reverse order starting with the programs last statement, the precondition of a statement is the postcondition of the next statement. Some statements like the while loop have a predetermined postcondition. In this case a verification condition is generated which states that the predetermined postcondition implies the precondition of the following statement. To

prove the correctness of a program every generated verification condition has to be proven. The following functions are used to calculate the verification conditions of statements:

$$\begin{aligned}
awp(\text{Empty}, Q, Q_R) &= Q \\
awp(\text{Assertion } a, Q, Q_R) &= a \\
awp(\text{ITE } c \text{ t } f, Q, Q_R) &= (awp(t, Q, Q_R) \wedge c) \vee (awp(f, Q, Q_R) \wedge \neg c) \\
awp(\text{While } c \text{ i } b, Q, Q_R) &= i \\
awp(\text{Seq } s1 \text{ } s2, Q, Q_R) &= awp(s1, awp(s2, Q, Q_R), Q_R) \\
awp(\text{Assignment } l \text{ e}, Q, Q_R) &= Q[update(\sigma, l, e)/\sigma] \\
awp(\text{Declaration } i, Q, Q_R) &= \text{simplify } Q \text{ with } i \neq \text{any other memory address} \\
awp(\text{FunCall } r \text{ fun args}, Q, Q_R) &= pre(\text{fun})[args/params(\text{fun})] \\
awp(\text{Return } r, Q, Q_R) &= Q_R[r/\backslash \text{result}]
\end{aligned}$$

$$\begin{aligned}
wvc(\text{Empty}, Q, Q_R) &= \emptyset \\
wvc(\text{Assertion } a, Q, Q_R) &= \{a \Rightarrow Q\} \\
wvc(\text{ITE } c \text{ t } f, Q, Q_R) &= wvc(t, Q, Q_R) \cup wvc(f, Q, Q_R) \\
wvc(\text{While } c \text{ i } b, Q, Q_R) &= \{i \wedge b \Rightarrow awp(b, i, Q_R), i \wedge \neg b \Rightarrow Q\} \cup wvc(b, i, Q_R) \\
wvc(\text{Seq } s1 \text{ } s2, Q, Q_R) &= wvc(s1, awp(s2, Q, Q_R), Q_R) \cup wvc(s2, Q, Q_R) \\
wvc(\text{Assignment } l \text{ e}, Q, Q_R) &= \emptyset \\
wvc(\text{Declaration } i, Q, Q_R) &= \emptyset \\
wvc(\text{FunCall } r \text{ fun args}, Q, Q_R) &= post(\text{fun})[args/params(\text{fun})][r/\backslash \text{result}] \Rightarrow Q \\
wvc(\text{Return } r, Q, Q_R) &= \emptyset
\end{aligned}$$

$Q[a/b]$ is the formula Q in which every occurrence of b has been replaced by a . If a and b are lists, a replacement is made for each element. $pre(f)$, $post(f)$, and $params(f)$ are the precondition, postcondition, and parameter list of the function f . The implementation of awp and wvc can be found in the module VC with some additions.

- The generated result of each awp call is simplified.
- Every time a formula, arithmetic expression, or LExpression is used to calculate the awp or wvc of a statement, its logic type is converted to

F0 and its memory model is converted to `Refs`. The GADTs that are used to model the different types of expressions ensure at compile time that the correct format is used.

- The generated verification are enriched with debug informations like line numbers.

These details are omitted in the definitions above to keep them small, as well as the line number which are part of some statements.

Based on the *wvc* and *awp* functions, verification conditions can be calculated separatly for each function of the program as follows:

$$\begin{aligned} wvcF(f) = & \{pre(f) \Rightarrow awp(body(f), false, post(f))\} \\ & \cup wvc(body(f), false, post(f)) \end{aligned}$$

The Floyd-Hoare logic rules implemented in `verify-c` follow the rules presented in the *Korrekte Software: Grundlagen und Methoden* course closely but the following changes were made:

- Declarations were added as separate statements to enable better simplification.
- Function calls produce a verification condition.
- The `\old(x)` syntax was dropped. Users are expected to bind values to logical variables in the precondition instead.

4.9 SMT Export

The generated verification conditions have to be proven in order to show the correctness of a program. This task is tedious. While the satisfiability of first order logic is not decideable in general, often formulas can be proven by a theorem prover. The `verify-c` exports the generated verification conditions into the SMT-LIB standard and runs the `Z3` solver in order to automate the proofs. While this feature is only rudimentarily implemented it already usable. In order to show how the export works Listing 12 shows one of the verification condition of the faculty program which was shown in section 3 in Listing 1. Exports of all verification conditions are located in the `.\target` folder created by `verify-c`.

Listing 12: an exported verification condition

```

1 (declare-fun read_array (Int Int) Int)
2 (declare-fun deref (Int) Int)
3 (declare-fun c () Int)
4 (declare-fun n () Int)
5 (declare-fun p () Int)
6 (declare-fun fac (Int) Int)
7 (assert (= (fac 0) 1))
8 (assert (forall ((nn Int)) (implies (< 0 nn) (= (fac nn) (* nn
    ↪ (fac (- nn 1)))))))
9 (assert (not (implies (and (and (and (= p (fac (- c 1))) (<= c
    ↪ (+ n 1))) (> c 0)) (<= c n)) (and (and (= (* p c) (fac
    ↪ (- (+ c 1) 1))) (<= (+ c 1) (+ n 1))) (> (+ c 1) 0))))))
10 (check-sat)

```

l.1 and l.2 of listing 12 are predefined functions which are contained in every export. `read_array` is a function which simulates array access. It takes two integer arguments. The first is interpreted as the memory address of an array, the second is an index. The return value is interpreted of the value of specified array at the specified index. `deref` is a function which simulates dereferencing. It takes one integer argument, which is interpreted as a memory address. The return value is interpreted as the value which is obtained by reading the memory at that address.

l.3 - l.5 are the unbound local variables which appear in the verification condition that should be exported. They are modelled as constant functions which do not take any input and return the value of that variable.

l.7 - l.8 contains the environment which is read from the optional `.env` accompanying the program which should be proven. The environment should contain custom functions and predicates as well as structure accessors.

l.9 contains the negated exported verification condition. In order to export the verification condition its memory model is converted back to **Plain**. For every boolean function which is supported by `verify-c` an equivalent SMT-LIB function exists, so they can be translated directly. The same goes for arithmetic functions except for the address operator. Verification conditions containing the address operator cannot be exported. The most difficult part is exporting the LExpressions. Identifiers are exported to variable names. Array access and dereferencing is modelled by the predefined functions men-

tioned above. Structure parts accessors are expected to be modelled by the user as part of the environment, where the name of the accessor should match the name of an unary function.

l.10 tells the smt solver to start the proof. If the solver can prove that there is no model for the negated verification condition (`unsat`), then the verification condition is valid and proven.

5 Limitations

`verify-c` is an educational project and should not be used to verify any critical software. It is only tested on small example programs. No guarantees are made by the author. This section should list known limitations without being exhaustive.

5.1 Correctness

While the purpose of the Floyd-Hoare logic is to prove correctness of programs, there are some classes errors which cannot be found by this implementation, including:

- nonterminating loops
- array index out of bounds errors
- overflows

5.2 Scalability

One important feature which is not implemented is the ability to specify which parts of the memory are changed by a function call. To remain sound without this specification only the postcondition of the called function can be assumed. This makes it hard (maybe even impossible) to prove programs like the recursive fibonacci program shown in listing 13.

Listing 13: a recursive fibonacci program

```
1 int fib(int n) {  
2     precondition("n >= 0");  
3     postcondition("\result == fibonacci(n)");  
4  
5     if(n == 0 || n == 1){  
6         return 1;  
7     } else {  
8         int prev = fib(n - 1);  
9         int prev2 = fib(n - 2);  
10        return prev + prev2;  
11    }  
12 }
```

This severely limits the scale in which `verify-c` can be applied.

5.3 Compatibility with C

`verify-c` only supports a subset of the C programming language. Missing features include:

- primitive types other than integers
- includes
- union types
- syntactic sugar like for loops or increment operators (`i++;`)